

XModAlg

Crossed Modules and Cat1-Algebras

1.18

17 November 2020

Zekeriya Arvasi

Alper Odabas

Zekeriya Arvasi

Email: zarvasi@ogu.edu.tr

Address: Prof. Dr. Z. Arvasi

Osmangazi University

Arts and Sciences Faculty

Department of Mathematics and Computer Science

Eskisehir

Turkey

Alper Odabas

Email: aodabas@ogu.edu.tr

Address: Dr. A. Odabas

Osmangazi University

Arts and Sciences Faculty

Department of Mathematics and Computer Science

Eskisehir

Turkey

Abstract

The XModAlg package provides functions for computation with crossed modules of commutative algebras and cat^1 -algebras.

Bug reports, suggestions and comments are, of course, welcome. Please submit an issue on GitHub at <http://github.com/gap-packages/xmodalg/issues/> or contact the second author at aodabas@ogu.edu.tr.

Copyright

© 2014-2020, Zekeriya Arvasi and Alper Odabas.

The XModAlg package is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version.

Acknowledgements

This documentation was prepared with the GAPDoc [LN17] and AutoDoc [GH17] packages.

The procedure used to produce new releases uses the package GitHubPagesForGAP [Hor17] and the package ReleaseTools.

Both authors are very grateful to Chris Wensley (<http://pages.bangor.ac.uk/~mas023/>) for helpful suggestions.

This work was partially supported by TÜBİTAK (The Scientific and Technical Research Council of Turkey), project number 107T542.

Contents

1	Introduction	4
2	Cat1-algebras	5
2.1	Definitions and examples	5
2.2	Cat ¹ –algebra morphisms	10
3	Crossed modules	13
3.1	Definition and Examples	13
3.2	(Pre-)Crossed Module Morphisms	17
4	Conversion between cat1-algebras and crossed modules	20
4.1	Equivalent Categories	20
	References	23
	Index	24

Chapter 1

Introduction

In 1950 S. MacLane and J.H.C. Whitehead, [Whi49] suggested that crossed modules modeled homotopy 2-types. Later crossed modules have been considered as *2-dimensional groups*, [Bro82], [Bro87]. The commutative algebra version of this construction has been adapted by T. Porter, [AP96], [Por87]. This algebraic version is called *combinatorial algebra theory*, which contains potentially important new ideas (see [AP96], [AP98], [AE03]).

A share package XMod, [AOUW17], [AW00], was prepared by M. Alp and C.D. Wensley for the GAP computational group theory language, initially for GAP3 then revised for GAP4. The 2-dimensional part of this programme contains functions for computing crossed modules and cat^1 -groups and their morphisms [AOUW17].

This package includes functions for computing crossed modules of algebras, cat^1 -algebras and their morphisms by analogy with *computational group theory*. We will concentrate on group rings over abelian groups over finite fields because these algebras are conveniently implemented in GAP. The tools needed are the group algebras in which the group algebra functor $\mathcal{K}(\cdot) : Gr \rightarrow Alg$ is left adjoint to the unit group functor $\mathcal{U}(\cdot) : Alg \rightarrow Gr$.

The categories XModAlg (crossed modules of algebras) and Cat1Alg (cat^1 -algebras) are equivalent, and we include functions to convert objects and morphisms between them. The algorithms implemented in this package are analyzed in A. Odabas's Ph.D. thesis, [Oda09] and described in detail in the paper [AO16].

There are aspects of commutative algebras for which no GAP functions yet exist, for example semidirect products. We have included here functions for all homomorphisms of algebras.

Chapter 2

Cat1-algebras

2.1 Definitions and examples

Algebraic structures which are equivalent to crossed modules of algebras include :

- cat^1 -algebras, (Ellis, [EII88]);
- simplicial algebras with Moore complex of length 1, (Z. Arvasi and T.Porter, [AP96]);
- algebra-algebroids, (Gaffar Musa's Ph.D. thesis, [Mos86]).

In this section we describe an implementation of cat^1 -algebras and their morphisms.

The notion of cat^1 -groups was defined as an algebraic model of 2-types by Loday in [Lod82]. Then Ellis defined the cat^1 -algebras in [EII88].

Let A and R be k -algebras, let $t, h : A \rightarrow R$ be surjections, and let $e : R \rightarrow A$ be an inclusion.

$$\begin{array}{c} A \\ \begin{array}{c} \curvearrowright \\ \parallel \\ \curvearrowleft \end{array} \\ \begin{array}{c} t \\ h \end{array} \\ R \end{array}$$

If the conditions,

$$\mathbf{Cat1Alg1} : te = id_R = he, \quad \mathbf{Cat1Alg2} : (\ker t)(\ker h) = \{0_A\}$$

are satisfied, then the algebraic system $\mathcal{C} := (e; t, h : A \rightarrow R)$ is called a *cat¹-algebra*. A system which satisfies the condition **Cat1Alg1** is called a *precat¹-algebra*. The homomorphisms t, h and e are called the *tail map*, *head map* and *range embedding* homomorphisms, respectively.

2.1.1 Cat1Algebra

- | | |
|--|-------------|
| ▷ Cat1Algebra(args) | (function) |
| ▷ PreCat1AlgebraByEndomorphisms(t, h) | (operation) |
| ▷ PreCat1AlgebraByTailHeadEmbedding(t, h, e) | (operation) |
| ▷ PreCat1Algebra(args) | (operation) |

- ▷ IsIdentityCat1Algebra(C) (property)
- ▷ IsCat1Algebra(C) (property)
- ▷ IsPreCat1Algebra(C) (property)

The operations listed above are used for construction of precat^1 - and cat^1 -algebra structures. The function `Cat1Algebra` selects the operation from the above implementations up to user's input. The operations `PreCat1AlgebraByEndomorphisms` and `PreCat1AlgebraByTailHeadEmbedding` are used with particular choices of algebra homomorphisms.

2.1.2 Source (for cat^1 -algebras)

- ▷ Source(C) (attribute)
- ▷ Range(C) (attribute)
- ▷ TailMap(C) (attribute)
- ▷ HeadMap(C) (attribute)
- ▷ RangeEmbedding(C) (attribute)
- ▷ Kernel(C) (method)
- ▷ Boundary(C) (attribute)

These are the seven main attributes of a pre-cat^1 -algebra.

Example

```
gap> Ac6 := GroupRing( GF(2), Group( (1,2,3)(4,5) ) );
gap> Rc3 := GroupRing( GF(2), Group( (1,2,3) ) );
gap> homAR := AllHomsOfAlgebras( Ac6, Rc3 );
gap> mgiAR := List( homAR, h -> MappingGeneratorsImages(h) );
gap> Print( mgiAR, "\n" );
[ [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ <zero> of ... ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ (Z(2)^0)*() ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ (Z(2)^0)*()+(Z(2)^0)*(1,2,3) ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ],
    [ (Z(2)^0)*()+(Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,3,2) ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ (Z(2)^0)*()+(Z(2)^0)*(1,3,2) ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ (Z(2)^0)*(1,2,3) ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ (Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,3,2) ] ],
  [ [ (Z(2)^0)*(1,3,2)(4,5) ], [ (Z(2)^0)*(1,3,2) ] ] ]
gap> homRA := AllHomsOfAlgebras( Rc3, Ac6 );
gap> mgiRA := List( homRA, h -> MappingGeneratorsImages(h) );
gap> Print( mgiRA, "\n" );
[ [ [ (Z(2)^0)*(1,2,3) ], [ <zero> of ... ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*() ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*()+(Z(2)^0)*(1,2,3) ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*()+(Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,3,2) ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*()+(Z(2)^0)*(1,3,2) ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*(1,2,3) ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,3,2) ] ],
  [ [ (Z(2)^0)*(1,2,3) ], [ (Z(2)^0)*(1,3,2) ] ] ]
gap> C4 := PreCat1AlgebraByTailHeadEmbedding( homAR[6], homAR[6], homRA[8] );
[AlgebraWithOne( GF(2), [ (Z(2)^0)*(1,2,3)(4,5) ] ) -> AlgebraWithOne( GF(2),
[ (Z(2)^0)*(1,2,3) ] )]
```

```

gap> IsCat1Algebra( C4 );
true
gap> Size( C4 );
[ 64, 8 ]
gap> Display( C4 );

Cat1-algebra [..=>..] :-
: source algebra has generators:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,2,3)(4,5) ]
: range algebra has generators:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,2,3) ]
: tail homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,3,2) ]
: head homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,3,2) ]
: range embedding maps range generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,3,2) ]
: kernel has generators:
  [ (Z(2)^0)*()+(Z(2)^0)*(4,5), (Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,2,3)(4,5),
    (Z(2)^0)*(1,3,2)+(Z(2)^0)*(1,3,2)(4,5) ]
: boundary homomorphism maps generators of kernel to:
  [ <zero> of ..., <zero> of ..., <zero> of ... ]
: kernel embedding maps generators of kernel to:
  [ (Z(2)^0)*()+(Z(2)^0)*(4,5), (Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,2,3)(4,5),
    (Z(2)^0)*(1,3,2)+(Z(2)^0)*(1,3,2)(4,5) ]

```

2.1.3 Cat1AlgebraSelect

▷ Cat1AlgebraSelect(n , $gpsize$, $gpnum$, num)

(operation)

The Cat1Algebra (2.1.1) function may also be used to select a cat^1 -algebra from a data file. All cat^1 -structures on commutative algebras are stored in a list in file cat1algdata.g. The data is read into the list CAT1ALG_LIST only when this function is called.

The function Cat1AlgebraSelect may be used in four ways:

- Cat1AlgebraSelect(n) returns the list of possible sizes of groups for group algebras with Galois field $GF(n)$.
- Cat1AlgebraSelect(n , m) returns the list of allowable group numbers with given Galois field $GF(n)$ and groups of size m .
- Cat1AlgebraSelect(n , m , k) returns the list of possible cat^1 -algebra structures with given Galois field $GF(n)$ and group number k of size m .
- Cat1AlgebraSelect(n , m , k , j) (or simply Cat1Algebra(n , m , k , j)) returns the j -th cat^1 -algebra structure with these other parameters.

Now, we give examples of the use of this function.

Example

```

gap> C := Cat1AlgebraSelect( 11 );

```

```

|-----|
| 11 is invalid number for Galois Field (GFnum) |
| Possible numbers for GFnum in the Data :      |
|-----|
[ 2, 3, 4, 5, 7 ]
Usage: Cat1Algebra( GFnum, gpsize, gpnum, num );
fail
gap> C := Cat1AlgebraSelect( 4, 12 );
|-----|
| 12 is invalid number for size of group (gpsize) |
| Possible numbers for the gpsize for GF(4) in the Data: |
|-----|
[ 1, 2, 3, 4, 5, 6, 7, 8, 9 ]
Usage: Cat1Algebra( GFnum, gpsize, gpnum, num );
fail
gap> C := Cat1AlgebraSelect( 2, 6, 3 );
|-----|
| 3 is invalid number for group of order 6 |
| Possible numbers for the gpnum in the Data : |
|-----|
[ 1, 2 ]
Usage: Cat1Algebra( GFnum, gpsize, gpnum, num );
fail
gap> C := Cat1AlgebraSelect( 2, 6, 2 );
There are 4 cat1-structures for the algebra GF(2)_c6.
  Range Alg      Tail      Head
|-----|
| GF(2)_c6      identity map      identity map |
| -----      [ 2, 10 ]          [ 2, 10 ] |
| -----      [ 2, 14 ]          [ 2, 14 ] |
| -----      [ 2, 50 ]          [ 2, 50 ] |
|-----|
Usage: Cat1Algebra( GFnum, gpsize, gpnum, num );
Algebra has generators [ (Z(2)^0)*(), (Z(2)^0)*(1,2,3)(4,5) ]
4
gap> C0 := Cat1AlgebraSelect( 4, 6, 2, 2 );
[GF(2^2)_c6 -> Algebra( GF(2^2),
[ (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)(2,5)(3,6)+
  Z(2)^0*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ] )]
gap> Size( C0 );
[ 4096, 1024 ]
gap> Display( C0 );

Cat1-algebra [GF(2^2)_c6=>..] :-
: source algebra has generators:
[ (Z(2)^0)*(), (Z(2)^0)*(1,2,3,4,5,6) ]
: range algebra has generators:
[ (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)(2,5)
  (3,6)+ (Z(2)^0)*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ]
: tail homomorphism maps source generators to:
[ (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)(2,5)
  (3,6)+ (Z(2)^0)*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ]

```



```

: head homomorphism maps source generators to:
[ (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)(2,5)
  (3,6)+ (Z(2)^0)*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ]
: range embedding maps range generators to:
[ (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)(2,5)
  (3,6)+ (Z(2)^0)*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ]
: kernel has generators:
[ (Z(2)^0)*()+ (Z(2)^0)*(1,2,3,4,5,6)+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)
  (2,5)(3,6)+ (Z(2)^0)*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ]
: boundary homomorphism maps generators of kernel to:
[ <zero> of ... ]
: kernel embedding maps generators of kernel to:
[ (Z(2)^0)*()+ (Z(2)^0)*(1,2,3,4,5,6)+ (Z(2)^0)*(1,3,5)(2,4,6)+ (Z(2)^0)*(1,4)
  (2,5)(3,6)+ (Z(2)^0)*(1,5,3)(2,6,4)+ (Z(2)^0)*(1,6,5,4,3,2) ]

```

2.1.4 SubCat1Algebra

- ▷ SubCat1Algebra(arg) (operation)
- ▷ SubPreCat1Algebra(arg) (operation)
- ▷ IsSubCat1Algebra(arg) (property)
- ▷ IsSubPreCat1Algebra(arg) (property)

Let $\mathcal{C} = (e; t, h : A \rightarrow R)$ be a cat^1 -algebra, and let A', R' be subalgebras of A and R respectively. If the restriction morphisms

$$t' = t|_{A'} : A' \rightarrow R', \quad h' = h|_{A'} : A' \rightarrow R', \quad e' = e|_{R'} : R' \rightarrow A'$$

satisfy the **Cat1Alg1** and **Cat1Alg2** conditions, then the system $\mathcal{C}' = (e'; t', h' : A' \rightarrow R')$ is called a *subcat¹-algebra* of $\mathcal{C} = (e; t, h : A \rightarrow R)$.

If the morphisms satisfy only the **Cat1Alg1** condition then \mathcal{C}' is called a *sub-precat¹-algebra* of \mathcal{C} .

The operations in this subsection are used for constructing subcat¹-algebras of a given cat^1 -algebra.

Example

```

gap> C3 := Cat1AlgebraSelect( 2, 6, 2, 4 );;
gap> A3 := Source( C3 );
GF(2)_c6
gap> B3 := Range( C3 );
GF(2)_c3
gap> eA3 := Elements( A3 );;
gap> eB3 := Elements( B3 );;
gap> AA3 := Subalgebra( A3, [ eA3[1], eA3[2], eA3[3] ] );
<algebra over GF(2), with 3 generators>
gap> [ Size(A3), Size(AA3) ];
[ 64, 4 ]
gap> BB3 := Subalgebra( B3, [ eB3[1], eB3[2] ] );
<algebra over GF(2), with 2 generators>
gap> [ Size(B3), Size(BB3) ];

```

```

[ 8, 2 ]
gap> CC3 := SubCat1Algebra( C3, AA3, BB3 );
[Algebra( GF(2), [ <zero> of ..., (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(4,5)
] ) -> Algebra( GF(2), [ <zero> of ..., (Z(2)^0)*() ] )]
gap> Display( CC3 );

Cat1-algebra [..=>..] :-
: source algebra has generators:
[ <zero> of ..., (Z(2)^0)*(), (Z(2)^0)*()+ (Z(2)^0)*(4,5) ]
: range algebra has generators:
[ <zero> of ..., (Z(2)^0)*() ]
: tail homomorphism maps source generators to:
[ <zero> of ..., (Z(2)^0)*(), <zero> of ... ]
: head homomorphism maps source generators to:
[ <zero> of ..., (Z(2)^0)*(), <zero> of ... ]
: range embedding maps range generators to:
[ <zero> of ..., (Z(2)^0)*() ]
: kernel has generators:
[ <zero> of ..., (Z(2)^0)*()+ (Z(2)^0)*(4,5) ]
: boundary homomorphism maps generators of kernel to:
[ <zero> of ..., <zero> of ... ]
: kernel embedding maps generators of kernel to:
[ <zero> of ..., (Z(2)^0)*()+ (Z(2)^0)*(4,5) ]

```

2.2 Cat¹–algebra morphisms

Let $\mathcal{C} = (e; t, h : A \rightarrow R)$, $\mathcal{C}' = (e'; t', h' : A' \rightarrow R')$ be cat¹-algebras, and let $\phi : A \rightarrow A'$ and $\varphi : R \rightarrow R'$ be algebra homomorphisms. If the diagram

$$\begin{array}{ccc}
 A & \xrightarrow{\phi} & A' \\
 \begin{array}{c} \curvearrowleft e \\ \parallel t \\ \parallel h \\ \curvearrowright \end{array} & & \begin{array}{c} \parallel t' \\ \parallel h' \\ \curvearrowright e' \end{array} \\
 R & \xrightarrow{\varphi} & R'
 \end{array}$$

commutes, (i.e. $t' \circ \phi = \varphi \circ t$, $h' \circ \phi = \varphi \circ h$ and $e' \circ \varphi = \phi \circ e$), then the pair (ϕ, φ) is called a cat¹-algebra morphism.

2.2.1 Cat1AlgebraMorphism

- ▷ Cat1AlgebraMorphism(arg) (operation)
- ▷ IdentityMapping(C) (method)
- ▷ PreCat1AlgebraMorphismByHoms(f, g) (operation)
- ▷ Cat1AlgebraMorphismByHoms(f, g) (operation)
- ▷ IsPreCat1AlgebraMorphism(C) (property)
- ▷ IsCat1AlgebraMorphism(arg) (property)

These operations are used for constructing cat^1 -algebra morphisms. Details of the implementations can be found in [Oda09].

2.2.2 Source (for morphisms of cat^1 -algebras)

▷ Source(m)	(attribute)
▷ Range(m)	(attribute)
▷ IsTotal(m)	(method)
▷ IsSingleValued(m)	(method)
▷ Name(m)	(method)
▷ Boundary(m)	(attribute)

These are the six main attributes of a cat^1 -algebra morphism.

Example

```
gap> C1 := Cat1Algebra( 2, 1, 1, 1 );
[GF(2)_triv -> GF(2)_triv]
gap> Display( C1 );

Cat1-algebra [GF(2)_triv=>GF(2)_triv] :-
: source algebra has generators:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: range algebra has generators:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: tail homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: head homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: range embedding maps range generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: the kernel is trivial.

gap> C2 := Cat1Algebra( 2, 2, 1, 2 );
[GF(2)_c2 -> GF(2)_triv]
gap> Display( C2 );

Cat1-algebra [GF(2)_c2=>GF(2)_triv] :-
: source algebra has generators:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,2) ]
: range algebra has generators:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: tail homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: head homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: range embedding maps range generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: kernel has generators:
  [ (Z(2)^0)*()+(Z(2)^0)*(1,2) ]
: boundary homomorphism maps generators of kernel to:
  [ <zero> of ... ]
```

```

: kernel embedding maps generators of kernel to:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,2) ]

gap> C1 = C2;
false
gap> R1 := Source( C1 );
gap> R2 := Source( C2 );
gap> S1 := Range( C1 );
gap> S2 := Range( C2 );
gap> gR1 := GeneratorsOfAlgebra( R1 );
[ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> gR2 := GeneratorsOfAlgebra( R2 );
[ (Z(2)^0)*(), (Z(2)^0)*(1,2) ]
gap> gS1 := GeneratorsOfAlgebra( S1 );
[ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> gS2 := GeneratorsOfAlgebra( S2 );
[ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> im1 := [ gR2[1], gR2[1] ];
[ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> f1 := AlgebraHomomorphismByImages( R1, R2, gR1, im1 );
[ (Z(2)^0)*(), (Z(2)^0)*() ] -> [ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> im2 := [ gS2[1], gS2[1] ];
[ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> f2 := AlgebraHomomorphismByImages( S1, S2, gS1, im2 );
[ (Z(2)^0)*(), (Z(2)^0)*() ] -> [ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> m := Cat1AlgebraMorphism( C1, C2, f1, f2 );
[[GF(2)_triv=>GF(2)_triv] => [GF(2)_c2=>GF(2)_triv]]
gap> Display( m );
Morphism of cat1-algebras :-
: Source = [GF(2)_triv=>GF(2)_triv] with generating sets:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: Range = [GF(2)_c2=>GF(2)_triv] with generating sets:
  [ (Z(2)^0)*(), (Z(2)^0)*(1,2) ]
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: Source Homomorphism maps source generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
: Range Homomorphism maps range generators to:
  [ (Z(2)^0)*(), (Z(2)^0)*() ]
gap> IsSurjective( m );
false
gap> IsInjective( m );
true
gap> IsBijective( m );
false

```

Chapter 3

Crossed modules

In this chapter we will present the notion of crossed modules of commutative algebras and their implementation in this package.

3.1 Definition and Examples

Let \mathbb{K} be a fixed commutative ring with $1 \neq 0$. From now on, all \mathbb{K} -algebras will be associative and commutative.

A *crossed module* is a \mathbb{K} -algebra morphism $\mathcal{X} := (\partial : S \rightarrow R)$ with an action of R on S satisfying

$$\mathbf{XModAlg\ 1} : \partial(r \cdot s) = r(\partial s), \quad \mathbf{XModAlg\ 2} : (\partial s) \cdot s' = ss',$$

for all $s, s' \in S$, $r \in R$. The morphism ∂ is called the *boundary map* of \mathcal{X} .

In this definition we used the left action notation. In the category of commutative algebras the right and the left actions coincide.

We can produce crossed modules by using the following methods.

3.1.1 XModAlgebra

▷ XModAlgebra(args)	(function)
▷ XModAlgebraByBoundaryAndAction(bdy, act)	(operation)
▷ XModAlgebraByIdeal(A, I)	(operation)
▷ XModAlgebraByModule(M, R)	(operation)
▷ XModAlgebraByCentralExtension(f)	(operation)
▷ XModAlgebraByMultipleAlgebra(A)	(operation)

Here are the standard constructions which these operations implement:

- Let A be an algebra and I an ideal of A . Then $\mathcal{X} = (inc : I \rightarrow A)$ is a crossed module with the multiplication action of A on I . Conversely, given a crossed module $\mathcal{X} = (\partial : S \rightarrow R)$, it is the case that $\partial(S)$ is an ideal of R .
- Let M be a R -module. Then $\mathcal{X} = (0 : M \rightarrow R)$ is a crossed module. Conversely, given a crossed module $\mathcal{X} = (\partial : M \rightarrow R)$, one can get that $\ker \partial$ is a $(R/\partial M)$ -module.

- Let $\partial : S \rightarrow R$ be a surjective algebra homomorphism. Define the action of R on S by $r \cdot s = \tilde{r}s$ where $\tilde{r} \in \partial^{-1}(r)$. Then $\mathcal{X} = (\partial : S \rightarrow R)$ is a crossed module with the defined action.
- Let S be a \mathbb{K} -algebra such that $\text{Ann}(S) = 0$ or $S^2 = S$. Then $\partial : S \rightarrow M(S)$ is a crossed module, where $M(S)$ is the algebra of multipliers of S and ∂ is the canonical homomorphism, [AE03].

3.1.2 Source (for crossed modules of commutative algebras)

▷ Source($X0$)	(attribute)
▷ Range($X0$)	(attribute)
▷ Boundary($X0$)	(attribute)
▷ XModAlgebraAction($X0$)	(attribute)

These four attributes are used in the construction of a crossed module \mathcal{X} where:

- Source(X) and Range(X) are the *source* and the *range* of the boundary map respectively;
- Boundary(X) is the boundary map of the crossed module \mathcal{X} ;
- XModAlgebraAction(X) is the action used in the crossed module.

The following standard GAP operations have special XModAlg implementations:

- Display(X) is used to list the components of \mathcal{X} ;
- Size(X) is used for calculating the order of the crossed module \mathcal{X} ;
- Name(X) is used for giving a name to the crossed module \mathcal{X} by associating the names of source and range algebras.

In the following example, we construct a crossed module by using the algebra GF_5D_4 and its augmentation ideal. We also show usage of the attributes listed above.

Example

```
gap> Ak4 := GroupRing( GF(5), DihedralGroup(4) );
<algebra-with-one over GF(5), with 2 generators>
gap> Size( Ak4 );
625
gap> SetName( Ak4, "GF5[k4]" );
gap> IAk4 := AugmentationIdeal( Ak4 );
<two-sided ideal in GF5[k4], (2 generators)>
gap> Size( IAk4 );
125
gap> SetName( IAk4, "I(GF5[k4])" );
gap> XIAk4 := XModAlgebraByIdeal( Ak4, IAk4 );
[ I(GF5[k4]) -> GF5[k4] ]
gap> Display( XIAk4 );

Crossed module [I(GF5[k4])->GF5[k4]] :-
: Source algebra I(GF5[k4]) has generators:
[ (Z(5)^2)*<identity> of ...+(Z(5)^0)*f1, (Z(5)^2)*<identity> of ...+(Z(5)^0)*f2 ]
```

```

: Range algebra GF5[k4] has generators:
[ (Z(5)^0)*<identity> of ..., (Z(5)^0)*f1, (Z(5)^0)*f2 ]
: Boundary homomorphism maps source generators to:
[ (Z(5)^2)*<identity> of ...+(Z(5)^0)*f1, (Z(5)^2)*<identity> of ...+(Z(5)^
0)*f2 ]

gap> Size( XIAk4 );
[ 125, 625 ]
gap> f := Boundary( XIAk4 );
MappingByFunction( I(GF5[k4]), GF5[k4], function( i ) ... end )
gap> Print( RepresentationsOfObject(XIAk4), "\n" );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsPreXModAlgebraObj" ]
gap> props := [ "CanEasilyCompareElements", "CanEasilySortElements",
> "IsDuplicateFree", "IsLeftActedOnByDivisionRing", "IsAdditivelyCommutative",
> "IsLDistributive", "IsRDistributive", "IsPreXModDomain", "Is2dAlgebraObject",
> "IsPreXModAlgebra", "IsXModAlgebra" ];
gap> known := KnownPropertiesOfObject( XIAk4 );
gap> ForAll( props, p -> (p in known) );
true
gap> Print( KnownAttributesOfObject(XIAk4), "\n" );
[ "Name", "Size", "Range", "Source", "Boundary", "XModAlgebraAction" ]

```

3.1.3 SubXModAlgebra

- ▷ SubXModAlgebra(X_0) (operation)
- ▷ IsSubXModAlgebra(X_0) (operation)

A crossed module $\mathcal{X}' = (\partial' : S' \rightarrow R')$ is a subcrossed module of the crossed module $\mathcal{X} = (\partial : S \rightarrow R)$ if $S' \leq S$, $R' \leq R$, $\partial' = \partial|_{S'}$, and the action of S' on R' is induced by the action of R on S . The operation SubXModAlgebra is used to construct a subcrossed module of a given crossed module.

Example

```

gap> e4 := Elements( IAK4 )[4];
(Z(5)^0)*<identity> of ...+(Z(5)^0)*f1+(Z(5)^2)*f2+(Z(5)^2)*f1*f2
gap> Je4 := Ideal( IAK4, [e4] );
gap> Size( Je4 );
5
gap> SetName( Je4, "<e4>" );
gap> XJe4 := XModAlgebraByIdeal( AK4, Je4 );
[ <e4> -> GF5[k4] ]
gap> Display( XJe4 );

Crossed module [<e4>->GF5[k4]] :-
: Source algebra <e4> has generators:
[ (Z(5)^0)*<identity> of ...+(Z(5)^0)*f1+(Z(5)^2)*f2+(Z(5)^2)*f1*f2 ]
: Range algebra GF5[k4] has generators:
[ (Z(5)^0)*<identity> of ..., (Z(5)^0)*f1, (Z(5)^0)*f2 ]
: Boundary homomorphism maps source generators to:
[ (Z(5)^0)*<identity> of ...+(Z(5)^0)*f1+(Z(5)^2)*f2+(Z(5)^2)*f1*f2 ]

```

```
gap> IsSubXModAlgebra( XIAk4, XJe4 );
true
```

3.1.4 PreXModAlgebraByBoundaryAndAction

- ▷ PreXModAlgebraByBoundaryAndAction(bdy, act) (operation)
- ▷ IsPreXModAlgebra(X0) (property)

An R -algebra homomorphism $\mathcal{X} := (\partial : S \rightarrow R)$ which satisfies the condition **XModAlg 1** is called a *precrossed module*. The details of these implementations can be found in [Oda09].

Example

```
gap> G := SmallGroup( 4, 2 );
<pc group of size 4 with 2 generators>
gap> F := GaloisField( 4 );
GF(2^2)
gap> R := GroupRing( F, G );
<algebra-with-one over GF(2^2), with 2 generators>
gap> Size( R );
256
gap> SetName( R, "GF(2^2)[k4]" );
gap> e5 := Elements( R )[5];
(Z(2)^0)*<identity> of ...+(Z(2)^0)*f1+(Z(2)^0)*f2+(Z(2)^0)*f1*f2
gap> S := Subalgebra( R, [e5] );
gap> SetName( S, "<e5>" );
gap> RS := Cartesian( R, S );
gap> SetName( RS, "GF(2^2)[k4] x <e5>" );
gap> act := AlgebraAction( R, RS, S );
gap> bdy := AlgebraHomomorphismByFunction( S, R, r->r );
MappingByFunction( <e5>, GF(2^2)[k4], function( r ) ... end )
gap> IsAlgebraAction( act );
true
gap> IsAlgebraHomomorphism( bdy );
true
gap> XM := PreXModAlgebraByBoundaryAndAction( bdy, act );
[<e5>->GF(2^2)[k4]]
gap> IsXModAlgebra( XM );
true
gap> Display( XM );

Crossed module [<e5>->GF(2^2)[k4]] :-
: Source algebra has generators:
  [ (Z(2)^0)*<identity> of ...+(Z(2)^0)*f1+(Z(2)^0)*f2+(Z(2)^0)*f1*f2 ]
: Range algebra GF(2^2)[k4] has generators:
  [ (Z(2)^0)*<identity> of ..., (Z(2)^0)*f1, (Z(2)^0)*f2 ]
: Boundary homomorphism maps source generators to:
  [ (Z(2)^0)*<identity> of ...+(Z(2)^0)*f1+(Z(2)^0)*f2+(Z(2)^0)*f1*f2 ]
```


3.2 (Pre-)Crossed Module Morphisms

Let $\mathcal{X} = (\partial : S \rightarrow R)$, $\mathcal{X}' = (\partial' : S' \rightarrow R')$ be (pre)crossed modules and $\theta : S \rightarrow S'$, $\varphi : R \rightarrow R'$ be algebra homomorphisms. If

$$\varphi \circ \partial = \partial' \circ \theta, \quad \theta(r \cdot s) = \varphi(r) \cdot \theta(s),$$

for all $r \in R, s \in S$, then the pair (θ, φ) is called a morphism between \mathcal{X} and \mathcal{X}' .

The conditions can be thought as the commutativity of the following diagrams:

$$\begin{array}{ccc} S & \xrightarrow{\theta} & S' \\ \partial \downarrow & & \downarrow \partial' \\ R & \xrightarrow{\varphi} & R' \end{array} \quad \begin{array}{ccc} R \times S & \xrightarrow{\varphi \times \theta} & R' \times S' \\ \downarrow & & \downarrow \\ S & \xrightarrow{\theta} & S'. \end{array}$$

In GAP we define the morphisms between algebraic structures such as cat^1 -algebras and crossed modules and they are investigated by the function `Make2dAlgebraMorphism`.

3.2.1 XModAlgebraMorphism

▷ <code>XModAlgebraMorphism(arg)</code>	(function)
▷ <code>IdentityMapping(X0)</code>	(method)
▷ <code>PreXModAlgebraMorphismByHoms(f, g)</code>	(operation)
▷ <code>XModAlgebraMorphismByHoms(f, g)</code>	(operation)
▷ <code>IsPreXModAlgebraMorphism(f)</code>	(property)
▷ <code>IsXModAlgebraMorphism(f)</code>	(property)
▷ <code>Source(m)</code>	(attribute)
▷ <code>Range(m)</code>	(attribute)
▷ <code>IsTotal(m)</code>	(method)
▷ <code>IsSingleValued(m)</code>	(method)
▷ <code>Name(m)</code>	(method)

These operations construct crossed module homomorphisms, which may have the attributes listed.

Example

```
gap> Ac4 := GroupRing( GF(2), CyclicGroup(4) );
<algebra-with-one over GF(2), with 2 generators>
gap> SetName( Ac4, "GF2[c4]" );
gap> IA4 := AugmentationIdeal( Ac4 );
<two-sided ideal in GF2[c4], (dimension 3)>
gap> SetName( IA4, "I(GF2[c4])" );
gap> XI4 := XModAlgebra( Ac4, IA4 );
[ I(GF2[c4]) -> GF2[c4] ]
gap> Bk4 := GroupRing( GF(2), SmallGroup( 4, 2 ) );
<algebra-with-one over GF(2), with 2 generators>
gap> SetName( Bk4, "GF2[k4]" );
gap> IBk4 := AugmentationIdeal( Bk4 );
<two-sided ideal in GF2[k4], (dimension 3)>
```

```

gap> SetName( IBk4, "I(GF2[k4])" );
gap> XIBk4 := XModAlgebra( Bk4, IBk4 );
[ I(GF2[k4]) -> GF2[k4] ]
gap> IAc4 = IBk4;
false
gap> homAB := AllHomsOfAlgebras( Ac4, Bk4 );;
gap> homIAIB := AllHomsOfAlgebras( IAc4, IBk4 );;
gap> mor := XModAlgebraMorphism( XIAC4, XIBk4, homIAIB[1], homAB[2] );
[[I(GF2[c4])->GF2[c4]] => [I(GF2[k4])->GF2[k4]]]
gap> Display( mor );

Morphism of crossed modules :-
: Source = [I(GF2[c4])->GF2[c4]]
: Range = [I(GF2[k4])->GF2[k4]]
: Source Homomorphism maps source generators to:
  [ <zero> of ..., <zero> of ..., <zero> of ... ]
: Range Homomorphism maps range generators to:
  [ (Z(2)^0)*<identity> of ..., (Z(2)^0)*<identity> of ...,
    (Z(2)^0)*<identity> of ... ]

gap> IsTotal( mor );
true
gap> IsSingleValued( mor );
true

```

3.2.2 Kernel (for morphisms of crossed modules of algebras)

▷ Kernel($X0$)

(method)

Let $(\theta, \varphi) : \mathcal{X} = (\partial : S \rightarrow R) \rightarrow \mathcal{X}' = (\partial' : S' \rightarrow R')$ be a crossed module homomorphism. Then the crossed module

$$\ker(\theta, \varphi) = (\partial| : \ker \theta \rightarrow \ker \varphi)$$

is called the *kernel* of (θ, φ) . Also, $\ker(\theta, \varphi)$ is an ideal of \mathcal{X} . An example is given below.

Example

```

gap> Xmor := Kernel( mor );
[ <algebra of dimension 3 over GF(2)> -> <algebra of dimension 3 over GF(2)> ]
gap> IsXModAlgebra( Xmor );
true
gap> Size( Xmor );
[ 8, 8 ]
gap> IsSubXModAlgebra( XIAC4, Xmor );
true

```

3.2.3 Image

▷ Image($X0$)

(operation)

Let $(\theta, \varphi) : \mathcal{X} = (\partial : S \rightarrow R) \rightarrow \mathcal{X}' = (\partial' : S' \rightarrow R')$ be a crossed module homomorphism. Then the crossed module

$$\mathfrak{I}(\theta, \varphi) = (\partial' | : \mathfrak{I}\theta \rightarrow \mathfrak{I}\varphi)$$

is called the image of (θ, φ) . Further, $\mathfrak{I}(\theta, \varphi)$ is a subcrossed module of (S', R', ∂') .

In this package, the image of a crossed module homomorphism can be obtained by the command `ImageSource`. The operation `Sub2dAlgObject` is effectively used for finding the kernel and image crossed modules induced from a given crossed module homomorphism.

3.2.4 SourceHom

▷ <code>SourceHom(m)</code>	(attribute)
▷ <code>RangeHom(m)</code>	(attribute)
▷ <code>IsInjective(m)</code>	(property)
▷ <code>IsSurjective(m)</code>	(property)
▷ <code>IsBijective(m)</code>	(property)

Let (θ, φ) be a homomorphism of crossed modules. If the homomorphisms θ and φ are injective (surjective) then (θ, φ) is injective (surjective).

The attributes `SourceHom` and `RangeHom` store the two algebra homomorphisms θ and φ .

Example

```
gap> ic4 := One( Ac4 );;
gap> theta := SourceHom( mor );;
gap> e1 := ic4*c4.1 + ic4*c4.2;
(Z(2)^0)*f1+(Z(2)^0)*f2
gap> ImageElm( theta, e1 );
<zero> of ...
gap> phi := RangeHom( mor );;
gap> e2 := ic4*c4.1;
(Z(2)^0)*f1
gap> ImageElm( phi, e2 );
(Z(2)^0)*<identity> of ...
gap> IsInjective( mor );
false
gap> IsSurjective( mor );
false
```

Chapter 4

Conversion between cat1-algebras and crossed modules

4.1 Equivalent Categories

The categories **Cat1Alg** (cat¹-algebras) and **XModAlg** (crossed modules) are naturally equivalent [EI88]. This equivalence is outlined in what follows. For a given crossed module $(\partial : A \rightarrow R)$ we can construct the semidirect product $R \ltimes A$ thanks to the action of R on A . If we define $t, h : R \ltimes A \rightarrow R$ and $e : R \rightarrow R \ltimes A$ by

$$t(r, a) = r, \quad h(r, a) = r + \partial(a), \quad e(r) = (r, 0),$$

respectively, then $\mathcal{C} = (e; t, h : R \ltimes A \rightarrow R)$ is a cat¹-algebra.

Conversely, for a given cat¹-algebra $\mathcal{C} = (e; t, h : A \rightarrow R)$, the map $\partial : \ker t \rightarrow R$ is a crossed module, where the action is multiplication action and ∂ is the restriction of h to $\ker t$.

Since all of these operations are linked to the functions `Cat1Algebra` (2.1.1) and `XModAlgebra` (3.1.1), they can be performed by calling these two functions. We may also use the function `Cat1Algebra` (2.1.1) instead of the operation `Cat1AlgebraSelect` (2.1.3).

4.1.1 Cat1AlgebraOfXModAlgebra

- ▷ `Cat1AlgebraOfXModAlgebra(X0)` (operation)
- ▷ `PreCat1AlgebraOfPreXModAlgebra(X0)` (operation)

These operations are used for constructing a cat¹-algebra from a given crossed module of algebras.

Example

```
gap> CXM := Cat1AlgebraOfXModAlgebra( XM );
[GF(2^2)[k4] IX <e5> -> GF(2^2)[k4]]
gap> Display( CXM );

Cat1-algebra [..=>GF(2^2)[k4]] :-
: range algebra has generators:
[ (Z(2)^0)*<identity> of ..., (Z(2)^0)*f1, (Z(2)^0)*f2 ]
: tail homomorphism maps source generators to:
: range embedding maps range generators to:
[ [ (Z(2)^0)*<identity> of ..., <zero> of ... ],
```

```

[ (Z(2)^0)*f1, <zero> of ... ], [ (Z(2)^0)*f2, <zero> of ... ] ]
: kernel has generators:
[ [ <zero> of ..., <zero> of ... ],
[ <zero> of ..., (Z(2)^0)*<identity> of ...+(Z(2)^0)*f1+(Z(2)^0)*f2+(Z(2)^0)*f1*f2 ],
[ <zero> of ..., (Z(2^2))*<identity> of ...+(Z(2^2))*f1+(Z(2^2))*f2+(Z(2^2))*f1*f2 ],
[ <zero> of ..., (Z(2^2)^2)*<identity> of ...+(Z(2^2)^2)*f1+(Z(2^2)^2)*f2+(Z(2^2)^2)*f1*f2 ] ]

```

4.1.2 XModAlgebraOfCat1Algebra

- ▷ XModAlgebraOfCat1Algebra(C) (operation)
- ▷ PreXModAlgebraOfPreCat1Algebra(C) (operation)

These operations are used for constructing a crossed module of algebras from a given cat^1 -algebra.

Example

```

gap> X3 := XModAlgebraOfCat1Algebra( C3 );
[ <algebra of dimension 3 over GF(2)> -> <algebra of dimension 3 over GF(2)> ]
gap> Display( X3 );

Crossed module [...->...] :-
: Source algebra has generators:
[ (Z(2)^0)*()+(Z(2)^0)*(4,5), (Z(2)^0)*(1,2,3)+(Z(2)^0)*(1,2,3)(4,5),
  (Z(2)^0)*(1,3,2)+(Z(2)^0)*(1,3,2)(4,5) ]
: Range algebra has generators:
[ (Z(2)^0)*(), (Z(2)^0)*(1,2,3), (Z(2)^0)*(1,3,2) ]
: Boundary homomorphism maps source generators to:
[ <zero> of ..., <zero> of ..., <zero> of ... ]

```

References

- [AE03] Z. Arvasi and U. Ege. Annihilators, multipliers and crossed modules. *Applied Categorical Structures*, 11:487–506, 2003. 4, 14
- [AO16] Z. Arvasi and A. Odabas. Computing 2-dimensional algebras: Crossed modules and cat1-algebras. *Journal of Algebra and Its Applications*, 15:1650185 (12 pages), 2016. 4
- [AOUW17] M. Alp, A. Odabas, E. O. Uslu, and C. D. Wensley. *XMod: Crossed Modules and Cat1-groups in GAP*, 2017. GAP package, <https://gap-packages.github.io/xmod/>. 4
- [AP96] Z. Arvasi and T. Porter. Simplicial and crossed resolutions of commutative algebras. *J. Algebra*, 181:426–448, 1996. 4, 5
- [AP98] Z. Arvasi and T. Porter. Freeness conditions for 2-crossed modules of commutative algebras. *Applied Categorical Structures*, 6:455–471, 1998. 4
- [AW00] M. Alp and C. D. Wensley. Enumeration of cat1-groups of low order. *Int. J. Algebra and Computation*, 10:407–424, 2000. 4
- [Bro82] R. Brown. Higher dimensional group theory. In *Low Dimensional Topology*, volume 48 of *London Math. Soc. Lecture Note Series*. London Mathematical Society, 1982. 4
- [Bro87] R. Brown. From groups to groupoids: a brief survey. *Bull. London Math. Soc.*, 19:113–134, 1987. 4
- [Ell88] G. J. Ellis. Higher dimensional crossed modules of algebras. *J. Pure and Appl. Algebra*, 52:277–282, 1988. 5, 20
- [GH17] S. Gutsche and M. Horn. *AutoDoc - Generate documentation from GAP source code (Version 2017.09.15)*, 2017. GAP package, <https://github.com/gap-packages/AutoDoc>. 2
- [Hor17] M. Horn. *GitHubPagesForGAP - a GitHub Pages generator for GAP packages*, 2017. GAP package, <https://gap-system.github.io/GitHubPagesForGAP/>. 2
- [LN17] F. Lübeck and M. Neunhöffer. *GAPDoc (version 1.6)*. RWTH Aachen, 2017. GAP package, <http://www.math.rwth-aachen.de/~Frank.Luebeck/GAPDoc/index.html>. 2
- [Lod82] J.-L. Loday. Spaces with finitely many non-trivial homotopy groups. *J. App. Algebra*, 24:179–202, 1982. 5

- [Mos86] G. H. Mosa. *Higher dimensional algebroids and crossed complexes*. Ph.d. thesis, University of Wales, Bangor (U.K.), 1986. [5](#)
- [Oda09] A. Odabas. *Crossed Modules of Algebras with GAP*. Ph.d. thesis, Osmangazi University, Eskisehir, 2009. [4](#), [11](#), [16](#)
- [Por87] T. Porter. Some categorical results in the theory of crossed modules in commutative algebras. *J. Algebra*, 109:415–429, 1987. [4](#)
- [Whi49] J. H. C. Whitehead. Combinatorial homotopy II. *Bull. Amer. Math. Soc.*, 55:453–496, 1949. [4](#)

Index

2d-algebra, [13](#)

Boundary

for cat1-algebras, [6](#)

for crossed modules of commutative algebras, [14](#)

for morphisms of cat1-algebras, [11](#)

cat1-group, [5](#)

Cat1Algebra, [5](#)

Cat1AlgebraMorphism, [10](#)

Cat1AlgebraMorphismByHoms, [10](#)

Cat1AlgebraOfXModAlgebra, [20](#)

Cat1AlgebraSelect, [7](#)

crossed module, [13](#)

HeadMap, [6](#)

IdentityMapping

for cat1-algebras, [10](#)

for crossed modules of algebras, [17](#)

Image, [18](#)

IsBijective, [19](#)

IsCat1Algebra, [6](#)

IsCat1AlgebraMorphism, [10](#)

IsIdentityCat1Algebra, [6](#)

IsInjective, [19](#)

IsPreCat1Algebra, [6](#)

IsPreCat1AlgebraMorphism, [10](#)

IsPreXModAlgebra, [16](#)

IsPreXModAlgebraMorphism, [17](#)

IsSingleValued

for morphisms of cat1-algebras, [11](#)

for morphisms of crossed modules of algebras, [17](#)

IsSubCat1Algebra, [9](#)

IsSubPreCat1Algebra, [9](#)

IsSubXModAlgebra, [15](#)

IsSurjective, [19](#)

IsTotal

for morphisms of cat1-algebras, [11](#)

for morphisms of crossed modules of algebras, [17](#)

IsXModAlgebraMorphism, [17](#)

Kernel

for cat1-algebras, [6](#)

for morphisms of crossed modules of algebras, [18](#)

Name

for morphisms of cat1-algebras, [11](#)

for morphisms of crossed modules of algebras, [17](#)

PreCat1Algebra, [5](#)

PreCat1AlgebraByEndomorphisms, [5](#)

PreCat1AlgebraByTailHeadEmbedding, [5](#)

PreCat1AlgebraMorphismByHoms, [10](#)

PreCat1AlgebraOfPreXModAlgebra, [20](#)

precrossed module, [16](#)

PreXModAlgebraByBoundaryAndAction, [16](#)

PreXModAlgebraMorphismByHoms, [17](#)

PreXModAlgebraOfPreCat1Algebra, [21](#)

Range

for cat1-algebras, [6](#)

for crossed modules of commutative algebras, [14](#)

for morphisms of cat1-algebras, [11](#)

for morphisms of crossed modules of algebras, [17](#)

RangeEmbedding, [6](#)

RangeHom, [19](#)

Source

for cat1-algebras, [6](#)

for crossed modules of commutative algebras, [14](#)

for morphisms of cat1-algebras, [11](#)

for morphisms of crossed modules of algebras, [17](#)
SourceHom, [19](#)
SubCat1Algebra, [9](#)
SubPreCat1Algebra, [9](#)
SubXModAlgebra, [15](#)

TailMap, [6](#)

XModAlgebra, [13](#)
XModAlgebraAction, [14](#)
XModAlgebraByBoundaryAndAction, [13](#)
XModAlgebraByCentralExtension, [13](#)
XModAlgebraByIdeal, [13](#)
XModAlgebraByModule, [13](#)
XModAlgebraByMultipleAlgebra, [13](#)
XModAlgebraMorphism, [17](#)
XModAlgebraMorphismByHoms, [17](#)
XModAlgebraOfCat1Algebra, [21](#)